Rules for integrands of the form $P_q[x]\ (a+b\,x^n)^p$

x: $\int P_q[x]\ (a+b\,x)^p\,dx$ when $p\in\mathbb{F}\ \land\ m+1\in\mathbb{Z}^-$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}^+$, then $F[x]\ (a+b\,x)^p = \frac{n}{b}\,\text{Subst}\left[x^{n\,p+n-1}\,F\left[-\frac{a}{b}+\frac{x^n}{b}\right],\,x,\,(a+b\,x)^{1/n}\right]\partial_x (a+b\,x)^{1/n}$

Rule: If $p\in\mathbb{F}\ \land\ m+1\in\mathbb{Z}^-$, let $n=\text{Denominator}[p]$, then

$$\int P_q[x]\ (a+b\,x)^p\,dx\ \longrightarrow\ \frac{n}{b}\,\text{Subst}\left[\int x^{n\,p+n-1}\,P_q\left[-\frac{a}{b}+\frac{x^n}{b}\right]dx,\,x,\,(a+b\,x)^{1/n}\right]$$

Program code:

```
(* Int[Pq_*(a_+b_.*x_)^p_,x_Symbol] :=
  With[{n=Denominator[p]},
  n/b*Subst[Int[x^(n*p+n-1)*ReplaceAll[Pq,x→-a/b+x^n/b],x],x,(a+b*x)^(1/n)]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && FractionQ[p] *)
```

2: $\int P_q[x]\ (a+b\,x^n)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int P_q[x]\ (a+b\,x^n)^p\,dx\ \longrightarrow\ \int \text{ExpandIntegrand}\left[P_q[x]\ (a+b\,x^n)^p,\,x\right]dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,n},x] && PolyQ[Pq,x] && (IGtQ[p,0] || EqQ[n,1])
```

3: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $P_q[x,0]=0$

Derivation: Algebraic simplification

Rule: If $P_q[x,0]=0$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\rightarrow\;\int x\,\text{PolynomialQuotient}[P_q[x],x,x]\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x^n)^p,x] /;
FreeQ[{a,b,n,p},x] && PolyQ[Pq,x] && EqQ[Coeff[Pq,x,0],0] && Not[MatchQ[Pq,x^m_.*u_. /; IntegerQ[m]]]
```

4. $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}$

1. $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+$

0: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\land q\geq n\land \text{PolynomialRemainder}[P_q[x],\,a+b\,x^n,\,x]==0$

Derivation: Algebraic simplification

Rule: If $n\in\mathbb{Z}^+\land q\geq n\land \text{PolynomialRemainder}[P_q[x],\,a+b\,x^n,\,x]==0$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\,\longrightarrow\,\int \text{PolynomialQuotient}[P_q[x],\,a+b\,x^n,\,x]\,(a+b\,x^n)^{p+1}\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[PolynomialQuotient[Pq,a+b*x^n,x]*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && IGtQ[n,0] && GeQ[Expon[Pq,x],n] && EqQ[PolynomialRemainder[Pq,a+b*x^n,x],0]
```

1: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}^+\land p>0$

Derivation: Binomial recurrence 1b applied q times

Rule: If $\frac{n-1}{2}\in\mathbb{Z}^+\land p>0$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\longrightarrow\;(a+b\,x^n)^p\sum_{i=0}^{q}\frac{P_q[x,\,i]\,x^{i+1}}{m+n\,p+i+1}+a\,n\,p\int(a+b\,x^n)^{p-1}\left(\sum_{i=0}^{q}\frac{P_q[x,\,i]\,x^i}{m+n\,p+i+1}\right)dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],i},
  (a+b*x^n)^p*Sum[Coeff[Pq,x,i]*x^(i+1)/(n*p+i+1),{i,0,q}] +
   a*n*p*Int[(a+b*x^n)^(p-1)*Sum[Coeff[Pq,x,i]*x^i/(n*p+i+1),{i,0,q}],x]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[(n-1)/2,0] && GtQ[p,0]
```

2. $\int P_q[x] \left(a+b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1$

   1. $\int P_q[x] \left(a+b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land q < n$

      1: $\int P_q[x] \left(a+b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land q == n-1$

Derivation: Algebraic expansion and binomial recurrence 2b applied $q-1$ times

Rule: If $n \in \mathbb{Z}^+ \land p < -1 \land q == n-1$, then

$$\int P_q[x] \left(a+b x^n\right)^p \, dx \longrightarrow$$
$$\frac{\left(a P_q[x,q] - b x \left(P_q[x] - P_q[x,q] \, x^q\right)\right) \left(a+b x^n\right)^{p+1}}{a b n (p+1)} + \frac{1}{a n (p+1)} \int \left(\sum_{i=0}^{q-1} (n (p+1)+i+1) \, P_q[x,i] \, x^i\right) \left(a+b x^n\right)^{p+1} dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],i},
  (a*Coeff[Pq,x,q]-b*x*ExpandToSum[Pq-Coeff[Pq,x,q]*x^q,x])*(a+b*x^n)^(p+1)/(a*b*n*(p+1)) +
  1/(a*n*(p+1))*Int[Sum[(n*(p+1)+i+1)*Coeff[Pq,x,i]*x^i,{i,0,q-1}]*(a+b*x^n)^(p+1),x] /;
 q==n-1] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[p,-1]
```

      2: $\int P_q[x] \left(a+b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land q < n-1$

Derivation: Binomial recurrence 2b applied q times

Note: $\sum_{i=0}^{q} (i+1) \, P_q[x,i] \, x^i == \partial_x \left(x \, P_q[x]\right)$ contributed by Martin Welz on 5 June 2015

Rule: If $n \in \mathbb{Z}^+ \land p < -1 \land q < n-1$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\longrightarrow$$

$$-\frac{x\,P_q[x]\,(a+b\,x^n)^{p+1}}{a\,n\,(p+1)}+\frac{1}{a\,n\,(p+1)}\int\left(\sum_{i=0}^{q}(n\,(p+1)+i+1)\,P_q[x,\,i]\,x^i\right)(a+b\,x^n)^{p+1}\,dx$$

$$-\frac{x\,P_q[x]\,(a+b\,x^n)^{p+1}}{a\,n\,(p+1)}+\frac{1}{a\,n\,(p+1)}\int(n\,(p+1)\,P_q[x]+\partial_x\,(x\,P_q[x]))\,(a+b\,x^n)^{p+1}\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  -x*Pq*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  1/(a*n*(p+1))*Int[ExpandToSum[n*(p+1)*Pq+D[x*Pq,x],x]*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[p,-1] && LtQ[Expon[Pq,x],n-1]
```

2. $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\land p<-1\land q\geq n$

1: $\int\frac{d+e\,x+f\,x^3+g\,x^4}{(a+b\,x^4)^{3/2}}\,dx$ when $b\,d+a\,g==0$

Rule: If $b\,d+a\,g==0$, then

$$\int\frac{d+e\,x+f\,x^3+g\,x^4}{(a+b\,x^4)^{3/2}}\,dx\;\longrightarrow\;-\frac{a\,f+2\,a\,g\,x-b\,e\,x^2}{2\,a\,b\,\sqrt{a+b\,x^4}}$$

Program code:

```
Int[P4_/(a_+b_.*x_^4)^(3/2),x_Symbol] :=
  With[{d=Coeff[P4,x,0],e=Coeff[P4,x,1],f=Coeff[P4,x,3],g=Coeff[P4,x,4]},
  -(a*f+2*a*g*x-b*e*x^2)/(2*a*b*Sqrt[a+b*x^4]) /;
 EqQ[b*d+a*g,0]] /;
FreeQ[{a,b},x] && PolyQ[P4,x,4] && EqQ[Coeff[P4,x,2],0]
```

2: $\int \frac{d + e x^2 + f x^3 + g x^4 + h x^6}{\left(a + b x^4\right)^{3/2}} \, dx$ when $b e - 3 a h == 0 \land b d + a g == 0$

Rule: If $b e - 3 a h == 0 \land b d + a g == 0$, then

$$\int \frac{d + e x^2 + f x^3 + g x^4 + h x^6}{\left(a + b x^4\right)^{3/2}} \, dx \longrightarrow -\frac{a f - 2 b d x - 2 a h x^3}{2 a b \sqrt{a + b x^4}}$$

Program code:

```
Int[P6_/(a_+b_.*x_^4)^(3/2),x_Symbol] :=
  With[{d=Coeff[P6,x,0],e=Coeff[P6,x,2],f=Coeff[P6,x,3],g=Coeff[P6,x,4],h=Coeff[P6,x,6]},
  -(a*f-2*b*d*x-2*a*h*x^3)/(2*a*b*Sqrt[a+b*x^4]) /;
 EqQ[b*e-3*a*h,0] && EqQ[b*d+a*g,0]] /;
FreeQ[{a,b},x] && PolyQ[P6,x,6] && EqQ[Coeff[P6,x,1],0] && EqQ[Coeff[P6,x,5],0]
```

3: $\int P_q[x] \left(a + b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land q \geq n$

Derivation: Algebraic expansion and binomial recurrence 2b applied $n - 1$ times

Note: $\sum_{i=0}^{q} (i + 1) P_q[x, i] x^i == \partial_x (x P_q[x])$ contributed by Martin Welz on 5 June 2015

Rule: If $n \in \mathbb{Z}^+ \land p < -1 \land q \geq n$, let $Q_{q-n}[x] = \text{PolynomialQuotient}[P_q[x], a + b x^n, x]$ and $R_{n-1}[x] = \text{PolynomialRemainder}[P_q[x], a + b x^n, x]$, then

$$\int P_q[x] \left(a + b x^n\right)^p \, dx \longrightarrow$$

$$\int R_{n-1}[x] \left(a + b x^n\right)^p \, dx + \int Q_{q-n}[x] \left(a + b x^n\right)^{p+1} \, dx \longrightarrow$$

$$-\frac{x\,R_{n-1}[x]\,\left(a+b\,x^n\right)^{p+1}}{a\,n\,(p+1)}+\frac{1}{a\,n\,(p+1)}\int\left(a\,n\,(p+1)\,Q_{q-n}[x]+n\,(p+1)\,R_{n-1}[x]+\partial_x\left(x\,R_{n-1}[x]\right)\right)\left(a+b\,x^n\right)^{p+1}\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Module[{Q=PolynomialQuotient[b^(Floor[(q-1)/n]+1)*Pq,a+b*x^n,x],
          R=PolynomialRemainder[b^(Floor[(q-1)/n]+1)*Pq,a+b*x^n,x]},
  -x*R*(a+b*x^n)^(p+1)/(a*n*(p+1)*b^(Floor[(q-1)/n]+1)) +
   1/(a*n*(p+1)*b^(Floor[(q-1)/n]+1))*Int[(a+b*x^n)^(p+1)*ExpandToSum[a*n*(p+1)*Q+n*(p+1)*R+D[x*R,x],x],x]] /;
 GeQ[q,n]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[p,-1]
```

3. $\int \frac{P_q[x]}{a + b x^n} \, dx$ when $n \in \mathbb{Z}^+ \land q < n$

1. $\int \frac{P_q[x]}{a + b x^3} \, dx$ when $n \in \mathbb{Z}^+ \land q < 3$

1. $\int \frac{A + B x}{a + b x^3} \, dx$

1: $\int \frac{A + B x}{a + b x^3} \, dx$ when $a B^3 - b A^3 == 0$

Derivation: Algebraic simplification

Basis: If $a B^3 - b A^3 == 0$, then $\frac{A+B x}{a+b x^3} == \frac{B^3}{b \left(A^2-A B x+B^2 x^2\right)}$

Rule: If $a B^3 - b A^3 == 0$, then

$$\int \frac{A + B x}{a + b x^3} \, dx \longrightarrow \frac{B^3}{b} \int \frac{1}{A^2 - A B x + B^2 x^2} \, dx$$

Program code:

```
Int[(A_+B_.*x_)/(a_+b_.*x_^3),x_Symbol] :=
  B^3/b*Int[1/(A^2-A*B*x+B^2*x^2),x] /;
FreeQ[{a,b,A,B},x] && EqQ[a*B^3-b*A^3,0]
```

2. $\int \frac{A+B\,x}{a+b\,x^3}\,dx$ when $a\,B^3-b\,A^3\neq 0$

1: $\int \frac{A+B\,x}{a+b\,x^3}\,dx$ when $a\,B^3-b\,A^3\neq 0\wedge\frac{a}{b}>0$

Reference: G&R 2.126.2, CRC 75

Derivation: Algebraic expansion

Basis: Let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/3}$, then $\frac{A+B\,x}{a+b\,x^3}==-\frac{r\,(B\,r-A\,s)}{3\,a\,s}\,\frac{1}{r+s\,x}+\frac{r}{3\,a\,s}\,\frac{r\,(B\,r+2\,A\,s)+s\,(B\,r-A\,s)\,x}{r^2-r\,s\,x+s^2\,x^2}$

Rule: If $a\,B^3-b\,A^3\neq 0\wedge\frac{a}{b}>0$, let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/3}$, then

$$\int\frac{A+B\,x}{a+b\,x^3}\,dx\to-\frac{r\,(B\,r-A\,s)}{3\,a\,s}\int\frac{1}{r+s\,x}\,dx+\frac{r}{3\,a\,s}\int\frac{r\,(B\,r+2\,A\,s)+s\,(B\,r-A\,s)\,x}{r^2-r\,s\,x+s^2\,x^2}\,dx$$

Program code:

```
Int[(A_+B_.*x_)/(a_+b_.*x_^3),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,3]], s=Denominator[Rt[a/b,3]]},
  -r*(B*r-A*s)/(3*a*s)*Int[1/(r+s*x),x] +
  r/(3*a*s)*Int[(r*(B*r+2*A*s)+s*(B*r-A*s)*x)/(r^2-r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b,A,B},x] && NeQ[a*B^3-b*A^3,0] && PosQ[a/b]
```

2: $\int \frac{A + B x}{a + b x^3} \, dx$ when $a B^3 - b A^3 \neq 0 \land \frac{a}{b} \not> 0$

Derivation: Algebraic expansion

Basis: Let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x}{a+b x^3} == \frac{r (B r+A s)}{3 a s (r-s x)} - \frac{r (r (B r-2 A s)-s (B r+A s) x)}{3 a s \left(r^2+r s x+s^2 x^2\right)}$

Rule: If $a B^3 - b A^3 \neq 0 \land \frac{a}{b} \not> 0$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{A + B x}{a + b x^3} \, dx \to \frac{r (B r + A s)}{3 a s} \int \frac{1}{r - s x} \, dx - \frac{r}{3 a s} \int \frac{r (B r - 2 A s) - s (B r + A s) x}{r^2 + r s x + s^2 x^2} \, dx$$

Program code:

```
Int[(A_+B_.*x_)/(a_+b_.*x_^3),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,3]], s=Denominator[Rt[-a/b,3]]},
  r*(B*r+A*s)/(3*a*s)*Int[1/(r-s*x),x] -
  r/(3*a*s)*Int[(r*(B*r-2*A*s)-s*(B*r+A*s)*x)/(r^2+r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b,A,B},x] && NeQ[a*B^3-b*A^3,0] && NegQ[a/b]
```

2. $\int \frac{A+B x+C x^2}{a+b x^3} dx$

1: $\int \frac{A+B x+C x^2}{a+b x^3} dx$ when $B^2 - A C == 0 \land b B^3 + a C^3 == 0$

Derivation: Algebraic simplification

Basis: If $B^2 - A C == 0 \land b B^3 + a C^3 == 0$, then $\frac{A+B x+C x^2}{a+b x^3} == -\frac{C^2}{b (B-C x)}$

Rule: If $B^2 - A C == 0 \land b B^3 + a C^3 == 0$, then

$$\int \frac{A+B x+C x^2}{a+b x^3} dx \longrightarrow -\frac{C^2}{b} \int \frac{1}{B-C x} dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    -C^2/b*Int[1/(B-C*x),x] /;
  EqQ[B^2-A*C,0] && EqQ[b*B^3+a*C^3,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

2. $\int \frac{A+B x+C x^2}{a+b x^3} dx$ when $A b^{2/3} - a^{1/3} b^{1/3} B - 2 a^{2/3} C == 0$

1: $\int \frac{A+B x+C x^2}{a+b x^3} dx$ when $A b^{2/3} - a^{1/3} b^{1/3} B - 2 a^{2/3} C == 0$

Derivation: Algebraic expansion

Basis: If $A b^{2/3} - a^{1/3} b^{1/3} B - 2 a^{2/3} C == 0$, let $q = \frac{a^{1/3}}{b^{1/3}}$, then $\frac{A+B x+C x^2}{a+b x^3} == \frac{C}{b (q+x)} + \frac{B+C q}{b (q^2-q x+x^2)}$

Rule: If $A b^{2/3} - a^{1/3} b^{1/3} B - 2 a^{2/3} C == 0$, let $q = \frac{a^{1/3}}{b^{1/3}}$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} dx \longrightarrow \frac{C}{b} \int \frac{1}{q + x} dx + \frac{B + C q}{b} \int \frac{1}{q^2 - q x + x^2} dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=a^(1/3)/b^(1/3)}, C/b*Int[1/(q+x),x] + (B+C*q)/b*Int[1/(q^2-q*x+x^2),x]] /;
  EqQ[A*b^(2/3)-a^(1/3)*b^(1/3)*B-2*a^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

2: $\int \frac{A + B x + C x^2}{a + b x^3} dx$ when $A (-b)^{2/3} - (-a)^{1/3} (-b)^{1/3} B - 2 (-a)^{2/3} C == 0$

Derivation: Algebraic expansion

Basis: If $A (-b)^{2/3} - (-a)^{1/3} (-b)^{1/3} B - 2 (-a)^{2/3} C == 0$, let $q = \frac{(-a)^{1/3}}{(-b)^{1/3}}$, then

$\frac{A+B x+C x^2}{a+b x^3} == \frac{C}{b (q+x)} + \frac{B+C q}{b (q^2-q x+x^2)}$

Rule: If $A (-b)^{2/3} - (-a)^{1/3} (-b)^{1/3} B - 2 (-a)^{2/3} C == 0$, let $q = \frac{(-a)^{1/3}}{(-b)^{1/3}}$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} dx \longrightarrow \frac{C}{b} \int \frac{1}{q + x} dx + \frac{B + C q}{b} \int \frac{1}{q^2 - q x + x^2} dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(-a)^(1/3)/(-b)^(1/3)}, C/b*Int[1/(q+x),x] + (B+C*q)/b*Int[1/(q^2-q*x+x^2),x]] /;
  EqQ[A*(-b)^(2/3)-(-a)^(1/3)*(-b)^(1/3)*B-2*(-a)^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

3: $\int \frac{A + B x + C x^2}{a + b x^3} \, dx$ when $A b^{2/3} + (-a)^{1/3} b^{1/3} B - 2 (-a)^{2/3} C == 0$

Derivation: Algebraic expansion

Basis: If $A b^{2/3} + (-a)^{1/3} b^{1/3} B - 2 (-a)^{2/3} C == 0$, let $q = \frac{(-a)^{1/3}}{b^{1/3}}$, then $\frac{A+B x+C x^2}{a+b x^3} == -\frac{C}{b (q-x)} + \frac{B-C q}{b \left(q^2+q x+x^2\right)}$

Rule: If $A b^{2/3} + (-a)^{1/3} b^{1/3} B - 2 (-a)^{2/3} C == 0$, let $q = \frac{(-a)^{1/3}}{b^{1/3}}$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} \, dx \longrightarrow -\frac{C}{b} \int \frac{1}{q - x} \, dx + \frac{B - C q}{b} \int \frac{1}{q^2 + q x + x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(-a)^(1/3)/b^(1/3)}, -C/b*Int[1/(q-x),x] + (B-C*q)/b*Int[1/(q^2+q*x+x^2),x]] /;
  EqQ[A*b^(2/3)+(-a)^(1/3)*b^(1/3)*B-2*(-a)^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

4: $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $A (-b)^{2/3}+a^{1/3} (-b)^{1/3} B-2 a^{2/3} C==0$

Derivation: Algebraic expansion

Basis: If $A (-b)^{2/3}+a^{1/3} (-b)^{1/3} B-2 a^{2/3} C==0$, let $q=\frac{a^{1/3}}{(-b)^{1/3}}$, then $\frac{A+B x+C x^2}{a+b x^3}==-\frac{C}{b (q-x)}+\frac{B-C q}{b \left(q^2+q x+x^2\right)}$

Rule: If $A (-b)^{2/3}+a^{1/3} (-b)^{1/3} B-2 a^{2/3} C==0$, let $q=\frac{a^{1/3}}{(-b)^{1/3}}$, then

$$\int \frac{A+B x+C x^2}{a+b x^3} \, dx \longrightarrow -\frac{C}{b} \int \frac{1}{q-x} \, dx+\frac{B-C q}{b} \int \frac{1}{q^2+q x+x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=a^(1/3)/(-b)^(1/3)}, -C/b*Int[1/(q-x),x] + (B-C*q)/b*Int[1/(q^2+q*x+x^2),x]] /;
  EqQ[A*(-b)^(2/3)+a^(1/3)*(-b)^(1/3)*B-2*a^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

5: $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $A-\left(\frac{a}{b}\right)^{1/3} B-2\left(\frac{a}{b}\right)^{2/3} C==0$

Derivation: Algebraic expansion

Basis: If $A-\left(\frac{a}{b}\right)^{1/3} B-2\left(\frac{a}{b}\right)^{2/3} C==0$, let $q=\left(\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3}==\frac{C}{b\,(q+x)}+\frac{B+C q}{b\left(q^2-q x+x^2\right)}$

Rule: If $A-\left(\frac{a}{b}\right)^{1/3} B-2\left(\frac{a}{b}\right)^{2/3} C==0$, let $q=\left(\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{A+B x+C x^2}{a+b x^3} \, dx \longrightarrow \frac{C}{b}\int \frac{1}{q+x} \, dx+\frac{B+C q}{b}\int \frac{1}{q^2-q x+x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(a/b)^(1/3)}, C/b*Int[1/(q+x),x] + (B+C*q)/b*Int[1/(q^2-q*x+x^2),x]] /;
  EqQ[A-(a/b)^(1/3)*B-2*(a/b)^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=Rt[a/b,3]}, C/b*Int[1/(q+x),x] + (B+C*q)/b*Int[1/(q^2-q*x+x^2),x]] /;
  EqQ[A-Rt[a/b,3]*B-2*Rt[a/b,3]^2*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

6: $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $A+\left(-\frac{a}{b}\right)^{1/3} B-2\left(-\frac{a}{b}\right)^{2/3} C==0$

Derivation: Algebraic expansion

Basis: If $A+\left(-\frac{a}{b}\right)^{1/3} B-2\left(-\frac{a}{b}\right)^{2/3} C==0$, let $q=\left(-\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3}==-\frac{C}{b\,(q-x)}+\frac{B-C q}{b\left(q^2+q x+x^2\right)}$

Rule: If $A+\left(-\frac{a}{b}\right)^{1/3} B-2\left(-\frac{a}{b}\right)^{2/3} C=0$, let $q=\left(-\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{A+B x+C x^2}{a+b x^3}\, dx \longrightarrow -\frac{C}{b} \int \frac{1}{q-x}\, dx+\frac{B-C q}{b} \int \frac{1}{q^2+q x+x^2}\, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(-a/b)^(1/3)}, -C/b*Int[1/(q-x),x] + (B-C*q)/b*Int[1/(q^2+q*x+x^2),x]] /;
  EqQ[A+(-a/b)^(1/3)*B-2*(-a/b)^(2/3)*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=Rt[-a/b,3]}, -C/b*Int[1/(q-x),x] + (B-C*q)/b*Int[1/(q^2+q*x+x^2),x]] /;
  EqQ[A+Rt[-a/b,3]*B-2*Rt[-a/b,3]^2*C,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

3: $\int \frac{A + B x + C x^2}{a + b x^3} \, dx$ when $a B^3 - b A^3 == 0 \lor \frac{a}{b} \notin \mathbb{Q}$

Derivation: Algebraic expansion

Basis: $\frac{A+B x+C x^2}{a+b x^3} == \frac{A+B x}{a+b x^3} + \frac{C x^2}{a+b x^3}$

Rule: If $a B^3 - b A^3 == 0 \lor \frac{a}{b} \notin \mathbb{Q}$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} \, dx \longrightarrow \int \frac{A + B x}{a + b x^3} \, dx + C \int \frac{x^2}{a + b x^3} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    Int[(A+B*x)/(a+b*x^3),x] + C*Int[x^2/(a+b*x^3),x] /;
  EqQ[a*B^3-b*A^3,0] || Not[RationalQ[a/b]]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

4. $\int \frac{A + B x + C x^2}{a + b x^3} \, dx$ when $A - B \left(\frac{a}{b}\right)^{1/3} + C \left(\frac{a}{b}\right)^{2/3} == 0$

1: $\int \frac{A + B x + C x^2}{a + b x^3} \, dx$ when $A - B \left(\frac{a}{b}\right)^{1/3} + C \left(\frac{a}{b}\right)^{2/3} == 0$

Derivation: Algebraic simplification

Basis: If $A - B \left(\frac{a}{b}\right)^{1/3} + C \left(\frac{a}{b}\right)^{2/3} == 0$, let $q = \left(\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3} == \frac{q^2}{a} \frac{A+C q x}{q^2-q x+x^2}$

Rule: If $A - B \left(\frac{a}{b}\right)^{1/3} + C \left(\frac{a}{b}\right)^{2/3} == 0$, let $q = \left(\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} \, dx \longrightarrow \frac{q^2}{a} \int \frac{A + C q x}{q^2 - q x + x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(a/b)^(1/3)}, q^2/a*Int[(A+C*q*x)/(q^2-q*x+x^2),x]] /;
  EqQ[A-B*(a/b)^(1/3)+C*(a/b)^(2/3),0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

2: $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $A+B\left(-\frac{a}{b}\right)^{1/3}+C\left(-\frac{a}{b}\right)^{2/3}==0$

Derivation: Algebraic simplification

Basis: If $A+B\left(-\frac{a}{b}\right)^{1/3}+C\left(-\frac{a}{b}\right)^{2/3}==0$, let $q=\left(-\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3}==\frac{q}{a}\frac{A q+(A+B q) x}{q^2+q x+x^2}$

Rule: If $A+B\left(-\frac{a}{b}\right)^{1/3}+C\left(-\frac{a}{b}\right)^{2/3}==0$, let $q=\left(-\frac{a}{b}\right)^{1/3}$, then

$$\int \frac{A+B x+C x^2}{a+b x^3} \, dx \rightarrow \frac{q}{a}\int \frac{A q+(A+B q) x}{q^2+q x+x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2]},
    With[{q=(-a/b)^(1/3)}, q/a*Int[(A*q+(A+B*q)*x)/(q^2+q*x+x^2),x]] /;
  EqQ[A+B*(-a/b)^(1/3)+C*(-a/b)^(2/3),0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2]
```

5. $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $A-B\left(\frac{a}{b}\right)^{1/3}+C\left(\frac{a}{b}\right)^{2/3}\neq 0$

1: $\int \frac{A+B x+C x^2}{a+b x^3} \, dx$ when $a B^3-b A^3\neq 0 \land \frac{a}{b}>0 \land A-B\left(\frac{a}{b}\right)^{1/3}+C\left(\frac{a}{b}\right)^{2/3}\neq 0$

Derivation: Algebraic expansion

Basis: Let $q=\left(\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3}==\frac{q\left(A-B q+C q^2\right)}{3 a (q+x)}+\frac{q\left(q\left(2 A+B q-C q^2\right)-\left(A-B q-2 C q^2\right) x\right)}{3 a\left(q^2-q x+x^2\right)}$

Rule: If $a B^3-b A^3\neq 0 \land \frac{a}{b}>0$, let $q=\left(\frac{a}{b}\right)^{1/3}$, if $A-B q+C q^2\neq 0$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} \, dx \longrightarrow \frac{q \left(A - B q + C q^2\right)}{3 a} \int \frac{1}{q + x} \, dx + \frac{q}{3 a} \int \frac{q \left(2 A + B q - C q^2\right) - \left(A - B q - 2 C q^2\right) x}{q^2 - q x + x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2],q=(a/b)^(1/3)},
    q*(A-B*q+C*q^2)/(3*a)*Int[1/(q+x),x] +
    q/(3*a)*Int[(q*(2*A+B*q-C*q^2)-(A-B*q-2*C*q^2)*x)/(q^2-q*x+x^2),x] /;
  NeQ[a*B^3-b*A^3,0] && NeQ[A-B*q+C*q^2,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2] && GtQ[a/b,0]
```

2: $\int \frac{A + B x + C x^2}{a + b x^3} \, dx$ when $a B^3 - b A^3 \neq 0 \land \frac{a}{b} < 0 \land A + B \left(-\frac{a}{b}\right)^{1/3} + C \left(-\frac{a}{b}\right)^{2/3} \neq 0$

Derivation: Algebraic expansion

Basis: Let $q = \left(-\frac{a}{b}\right)^{1/3}$, then $\frac{A+B x+C x^2}{a+b x^3} == \frac{q \left(A+B q+C q^2\right)}{3 a (q-x)} + \frac{q \left(q \left(2 A-B q-C q^2\right)+\left(A+B q-2 C q^2\right) x\right)}{3 a \left(q^2+q x+x^2\right)}$

Rule: If $a B^3 - b A^3 \neq 0 \land \frac{a}{b} < 0$, let $q = \left(-\frac{a}{b}\right)^{1/3}$, if $A + B q + C q^2 \neq 0$, then

$$\int \frac{A + B x + C x^2}{a + b x^3} \, dx \longrightarrow \frac{q \left(A + B q + C q^2\right)}{3 a} \int \frac{1}{q - x} \, dx + \frac{q}{3 a} \int \frac{q \left(2 A - B q - C q^2\right) + \left(A + B q - 2 C q^2\right) x}{q^2 + q x + x^2} \, dx$$

Program code:

```
Int[P2_/(a_+b_.*x_^3),x_Symbol] :=
  With[{A=Coeff[P2,x,0],B=Coeff[P2,x,1],C=Coeff[P2,x,2],q=(-a/b)^(1/3)},
    q*(A+B*q+C*q^2)/(3*a)*Int[1/(q-x),x] +
    q/(3*a)*Int[(q*(2*A-B*q-C*q^2)+(A+B*q-2*C*q^2)*x)/(q^2+q*x+x^2),x] /;
  NeQ[a*B^3-b*A^3,0] && NeQ[A+B*q+C*q^2,0]] /;
FreeQ[{a,b},x] && PolyQ[P2,x,2] && LtQ[a/b,0]
```

2: $\int \frac{P_q[x]}{a + b\,x^n}\,dx$ when $\frac{n}{2} \in \mathbb{Z}^+ \wedge q < n$

Derivation: Algebraic expansion

Basis: If $\frac{n}{2} \in \mathbb{Z} \wedge q < n$, then $P_q[x] = \sum_{i=0}^{n-1} x^i\,P_q[x,\,i] = \sum_{i=0}^{n/2-1} x^i\,\left(P_q[x,\,i] + P_q\left[x,\,\frac{n}{2} + i\right]\,x^{n/2}\right)$

Note: The resulting integrands are of the form $\frac{x^q\,\left(r+s\,x^{n/2}\right)}{a+b\,x^n}$ for which there are rules.

Rule: If $\frac{n}{2} \in \mathbb{Z}^+ \wedge q < n$, then

$$\int \frac{P_q[x]}{a + b\,x^n}\,dx \longrightarrow \int \sum_{i=0}^{n/2-1} \frac{x^i\,\left(P_q[x,\,i] + P_q\left[x,\,\frac{n}{2} + i\right]\,x^{n/2}\right)}{c^i\,\left(a + b\,x^n\right)}\,dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^n_),x_Symbol] :=
  With[{v=Sum[x^ii*(Coeff[Pq,x,ii]+Coeff[Pq,x,n/2+ii]*x^(n/2))/(a+b*x^n),{ii,0,n/2-1}]},
  Int[v,x] /;
 SumQ[v]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n/2,0] && Expon[Pq,x]<n
```

4. $\int \frac{P_q[x]}{\sqrt{a+b\,x^n}}\,dx$ when $n \in \mathbb{Z}^+ \land q < n-1$

1. $\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx$

1. $\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx$ when $a > 0$

1: $\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx$ when $a > 0 \land b\,c^3 - 2\left(5 - 3\sqrt{3}\right) a\,d^3 == 0$

Reference: G&R 3.139

Note: If $a > 0 \land b > 0$, then $\text{ArcSin}\left[\frac{-1+\sqrt{3}-\left(\frac{b}{a}\right)^{1/3} x}{1+\sqrt{3}+\left(\frac{b}{a}\right)^{1/3} x}\right]$ is real when $\sqrt{a+b\,x^3}$ is real.

Warning: The result is discontinuous on the real line when $x == -\frac{1+\sqrt{3}}{q}$ where $q \to \left(\frac{b}{a}\right)^{1/3}$.

Rule: If $a > 0 \land b\,c^3 - 2\left(5 - 3\sqrt{3}\right) a\,d^3 == 0$, let $q \to \frac{r}{s} \to \frac{\left(1-\sqrt{3}\right) d}{c}$, then

$$\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx \longrightarrow \frac{2\,d\sqrt{a+b\,x^3}}{a\,q^2\left(1+\sqrt{3}+q\,x\right)} + \frac{3^{1/4}\sqrt{2-\sqrt{3}}\;d\,(1+q\,x)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1+\sqrt{3}+q\,x\right)^2}}}{q^2\sqrt{a+b\,x^3}\sqrt{\frac{1+q\,x}{\left(1+\sqrt{3}+q\,x\right)^2}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{-1+\sqrt{3}-q\,x}{1+\sqrt{3}+q\,x}\right], -7-4\sqrt{3}\right]$$

$$\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx \longrightarrow \frac{2\,d\,s^3\sqrt{a+b\,x^3}}{a\,r^2\left(\left(1+\sqrt{3}\right)s+r\,x\right)} - \frac{3^{1/4}\sqrt{2-\sqrt{3}}\;d\,s\,(s+r\,x)\sqrt{\frac{s^2-r\,s\,x+r^2\,x^2}{\left(\left(1+\sqrt{3}\right)s+r\,x\right)^2}}}{r^2\sqrt{a+b\,x^3}\sqrt{\frac{s\,(s+r\,x)}{\left(\left(1+\sqrt{3}\right)s+r\,x\right)^2}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\left(1-\sqrt{3}\right)s+r\,x}{\left(1+\sqrt{3}\right)s+r\,x}\right], -7-4\sqrt{3}\right]$$

Program code:

```
Int[(c_+d_.*x_)/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Simplify[(1-Sqrt[3])*d/c]], s=Denom[Simplify[(1-Sqrt[3])*d/c]]},
  2*d*s^3*Sqrt[a+b*x^3]/(a*r^2*((1+Sqrt[3])*s+r*x)) -
  3^(1/4)*Sqrt[2-Sqrt[3]]*d*s*(s+r*x)*Sqrt[(s^2-r*s*x+r^2*x^2)/((1+Sqrt[3])*s+r*x)^2]/
    (r^2*Sqrt[a+b*x^3]*Sqrt[s*(s+r*x)/((1+Sqrt[3])*s+r*x)^2])*
    EllipticE[ArcSin[((1-Sqrt[3])*s+r*x)/((1+Sqrt[3])*s+r*x)],-7-4*Sqrt[3]]] /;
FreeQ[{a,b,c,d},x] && PosQ[a] && EqQ[b*c^3-2*(5-3*Sqrt[3])*a*d^3,0]
```

2: $\int \frac{c+d x}{\sqrt{a+b x^3}} \, dx$ when $a>0 \land b c^3-2\left(5-3\sqrt{3}\right) a d^3 \neq 0$

Derivation: Algebraic expansion

Note: Second integrand is of the form $\frac{c+d x}{\sqrt{a+b x^3}}$ where $a>0 \land b c^3-2\left(5-3\sqrt{3}\right) a d^3 == 0$.

Rule: If $a>0 \land b c^3-2\left(5-3\sqrt{3}\right) a d^3 \neq 0$, let $\frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{c+d x}{\sqrt{a+b x^3}} \, dx \to \frac{c r-\left(1-\sqrt{3}\right) d s}{r} \int \frac{1}{\sqrt{a+b x^3}} \, dx+\frac{d}{r} \int \frac{\left(1-\sqrt{3}\right) s+r x}{\sqrt{a+b x^3}} \, dx$$

Program code:

```
Int[(c_+d_.*x_)/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  (c*r-(1-Sqrt[3])*d*s)/r*Int[1/Sqrt[a+b*x^3],x] + d/r*Int[((1-Sqrt[3])*s+r*x)/Sqrt[a+b*x^3],x]] /;
FreeQ[{a,b,c,d},x] && PosQ[a] && NeQ[b*c^3-2*(5-3*Sqrt[3])*a*d^3,0]
```

2. $\int \frac{c+d x}{\sqrt{a+b x^3}} \, dx$ when $a \ngtr 0$

1: $\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx$ when $a \not> 0 \land b\,c^3 - 2\left(5+3\sqrt{3}\right) a\,d^3 == 0$

Reference: G&R 3.139

Note: If $a < 0 \land b < 0$, then $\text{ArcSin}\left[\frac{1+\sqrt{3}+\left(\frac{b}{a}\right)^{1/3} x}{-1+\sqrt{3}-\left(\frac{b}{a}\right)^{1/3} x}\right]$ is real when $\sqrt{a+b\,x^3}$ is real.

Warning: The result is discontinuous on the real line when $x == -\frac{1-\sqrt{3}}{q}$ where $q \to \left(\frac{b}{a}\right)^{1/3}$.

Rule: If $a \not> 0 \land b\,c^3 - 2\left(5+3\sqrt{3}\right) a\,d^3 == 0$, let $q \to \frac{r}{s} \to \frac{\left(1+\sqrt{3}\right) d}{c}$, then

$$\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx \to \frac{2\,d\,\sqrt{a+b\,x^3}}{a\,q^2\left(1-\sqrt{3}+q\,x\right)} + \frac{3^{1/4}\sqrt{2+\sqrt{3}}\;d\,(1+q\,x)\sqrt{\frac{1-q\,x+q^2\,x^2}{\left(1-\sqrt{3}+q\,x\right)^2}}}{q^2\sqrt{a+b\,x^3}\sqrt{-\frac{1+q\,x}{\left(1-\sqrt{3}+q\,x\right)^2}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{1+\sqrt{3}+q\,x}{1-\sqrt{3}+q\,x}\right],\,-7+4\sqrt{3}\right]$$

$$\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx \to \frac{2\,d\,s^3\sqrt{a+b\,x^3}}{a\,r^2\left(\left(1-\sqrt{3}\right)s+r\,x\right)} + \frac{3^{1/4}\sqrt{2+\sqrt{3}}\;d\,s\,(s+r\,x)\sqrt{\frac{s^2-r\,s\,x+r^2\,x^2}{\left(\left(1-\sqrt{3}\right)s+r\,x\right)^2}}}{r^2\sqrt{a+b\,x^3}\sqrt{-\frac{s\,(s+r\,x)}{\left(\left(1-\sqrt{3}\right)s+r\,x\right)^2}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{\left(1+\sqrt{3}\right)s+r\,x}{\left(1-\sqrt{3}\right)s+r\,x}\right],\,-7+4\sqrt{3}\right]$$

Program code:

```
Int[(c_+d_.*x_)/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Simplify[(1+Sqrt[3])*d/c]], s=Denom[Simplify[(1+Sqrt[3])*d/c]]},
  2*d*s^3*Sqrt[a+b*x^3]/(a*r^2*((1-Sqrt[3])*s+r*x)) +
  3^(1/4)*Sqrt[2+Sqrt[3]]*d*s*(s+r*x)*Sqrt[(s^2-r*s*x+r^2*x^2)/((1-Sqrt[3])*s+r*x)^2]/
    (r^2*Sqrt[a+b*x^3]*Sqrt[-s*(s+r*x)/((1-Sqrt[3])*s+r*x)^2])*
    EllipticE[ArcSin[((1+Sqrt[3])*s+r*x)/((1-Sqrt[3])*s+r*x)],-7+4*Sqrt[3]]] /;
FreeQ[{a,b,c,d},x] && NegQ[a] && EqQ[b*c^3-2*(5+3*Sqrt[3])*a*d^3,0]
```

2: $\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx$ when $a \ngtr 0\ \land\ b\,c^3-2\left(5+3\sqrt{3}\right)a\,d^3 \neq 0$

Derivation: Algebraic expansion

Note: Second integrand is of the form $\frac{c+d\,x}{\sqrt{a+b\,x^3}}$ where $a \ngtr 0\ \land\ b\,c^3-2\left(5+3\sqrt{3}\right)a\,d^3 == 0$.

Rule: If $a \ngtr 0\ \land\ b\,c^3-2\left(5+3\sqrt{3}\right)a\,d^3 \neq 0$, let $q \to \frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{c+d\,x}{\sqrt{a+b\,x^3}}\,dx \longrightarrow \frac{c\,r-\left(1+\sqrt{3}\right)d\,s}{r}\int \frac{1}{\sqrt{a+b\,x^3}}\,dx+\frac{d}{r}\int \frac{\left(1+\sqrt{3}\right)s+r\,x}{\sqrt{a+b\,x^3}}\,dx$$

Program code:

```
Int[(c_+d_.*x_)/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  (c*r-(1+Sqrt[3])*d*s)/r*Int[1/Sqrt[a+b*x^3],x] + d/r*Int[((1+Sqrt[3])*s+r*x)/Sqrt[a+b*x^3],x]] /;
FreeQ[{a,b,c,d},x] && NegQ[a] && NeQ[b*c^3-2*(5+3*Sqrt[3])*a*d^3,0]
```

2. $\int \frac{c+d\,x^4}{\sqrt{a+b\,x^6}}\,dx$

1: $\int \frac{c+d\,x^4}{\sqrt{a+b\,x^6}}\,dx$ when $2\left(\frac{b}{a}\right)^{2/3}c-\left(1-\sqrt{3}\right)d == 0$

Rule: If $2\left(\frac{b}{a}\right)^{2/3}c-\left(1-\sqrt{3}\right)d == 0$, let $\frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{c+d\,x^4}{\sqrt{a+b\,x^6}}\,dx \longrightarrow$$

$$\frac{\left(1+\sqrt{3}\right) d\, s^3\, x\, \sqrt{a+b\, x^6}}{2\, a\, r^2\, \left(s+\left(1+\sqrt{3}\right) r\, x^2\right)}-\frac{3^{1/4}\, d\, s\, x\, \left(s+r\, x^2\right) \sqrt{\frac{s^2-r\, s\, x^2+r^2\, x^4}{\left(s+\left(1+\sqrt{3}\right) r\, x^2\right)^2}}}{2\, r^2\, \sqrt{\frac{r\, x^2\, \left(s+r\, x^2\right)}{\left(s+\left(1+\sqrt{3}\right) r\, x^2\right)^2}}\, \sqrt{a+b\, x^6}}\, \text{EllipticE}\left[\text{ArcCos}\left[\frac{s+\left(1-\sqrt{3}\right) r\, x^2}{s+\left(1+\sqrt{3}\right) r\, x^2}\right],\, \frac{2+\sqrt{3}}{4}\right]$$

Program code:

```
Int[(c_+d_.*x_^4)/Sqrt[a_+b_.*x_^6],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  (1+Sqrt[3])*d*s^3*x*Sqrt[a+b*x^6]/(2*a*r^2*(s+(1+Sqrt[3])*r*x^2)) -
  3^(1/4)*d*s*x*(s+r*x^2)*Sqrt[(s^2-r*s*x^2+r^2*x^4)/(s+(1+Sqrt[3])*r*x^2)^2]/
    (2*r^2*Sqrt[(r*x^2*(s+r*x^2))/(s+(1+Sqrt[3])*r*x^2)^2]*Sqrt[a+b*x^6])*
    EllipticE[ArcCos[(s+(1-Sqrt[3])*r*x^2)/(s+(1+Sqrt[3])*r*x^2)],(2+Sqrt[3])/4]] /;
FreeQ[{a,b,c,d},x] && EqQ[2*Rt[b/a,3]^2*c-(1-Sqrt[3])*d,0]
```

2: $\int \frac{c+d x^4}{\sqrt{a+b x^6}} \, dx$ when $2 \left(\frac{b}{a}\right)^{2/3} c-\left(1-\sqrt{3}\right) d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{c+d x^4}{\sqrt{a+b x^6}} = \frac{2 c q^2-\left(1-\sqrt{3}\right) d}{2 q^2 \sqrt{a+b x^6}} + \frac{d \left(1-\sqrt{3}+2 q^2 x^4\right)}{2 q^2 \sqrt{a+b x^6}}$

Rule: If $2 \left(\frac{b}{a}\right)^{2/3} c-\left(1-\sqrt{3}\right) d \neq 0$, let $q=\left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{c+d x^4}{\sqrt{a+b x^6}} \, dx \rightarrow \frac{2 c q^2-\left(1-\sqrt{3}\right) d}{2 q^2} \int \frac{1}{\sqrt{a+b x^6}} \, dx + \frac{d}{2 q^2} \int \frac{1-\sqrt{3}+2 q^2 x^4}{\sqrt{a+b x^6}} \, dx$$

Program code:

```
Int[(c_+d_.*x_^4)/Sqrt[a_+b_.*x_^6],x_Symbol] :=
  With[{q=Rt[b/a,3]},
  (2*c*q^2-(1-Sqrt[3])*d)/(2*q^2)*Int[1/Sqrt[a+b*x^6],x] + d/(2*q^2)*Int[(1-Sqrt[3]+2*q^2*x^4)/Sqrt[a+b*x^6],x]] /;
FreeQ[{a,b,c,d},x] && NeQ[2*Rt[b/a,3]^2*c-(1-Sqrt[3])*d,0]
```

3. $\int \frac{c+d x^2}{\sqrt{a+b x^8}} \, dx$

1: $\int \frac{c+d x^2}{\sqrt{a+b x^8}} \, dx$ when $b c^4-a d^4 = 0$

Rule: If $b c^4-a d^4 = 0$, then

$$\int \frac{c+d x^2}{\sqrt{a+b x^8}} \, dx \rightarrow$$

$$-\frac{c\, d\, x^3 \sqrt{-\frac{(c-d\,x^2)^2}{c\,d\,x^2}}\ \sqrt{-\frac{d^2\,(a+b\,x^8)}{b\,c^2\,x^4}}}{\sqrt{2+\sqrt{2}}\ (c-d\,x^2)\ \sqrt{a+b\,x^8}}\ \text{EllipticF}\left[\text{ArcSin}\left[\frac{1}{2}\sqrt{\frac{\sqrt{2}\ c^2+2\,c\,d\,x^2+\sqrt{2}\ d^2\,x^4}{c\,d\,x^2}}\right],\ -2\left(1-\sqrt{2}\right)\right]$$

Program code:

```
Int[(c_+d_.*x_^2)/Sqrt[a_+b_.*x_^8],x_Symbol] :=
  -c*d*x^3*Sqrt[-(c-d*x^2)^2/(c*d*x^2)]*Sqrt[-d^2*(a+b*x^8)/(b*c^2*x^4)]/(Sqrt[2+Sqrt[2]]*(c-d*x^2)*Sqrt[a+b*x^8])*
    EllipticF[ArcSin[1/2*Sqrt[(Sqrt[2]*c^2+2*c*d*x^2+Sqrt[2]*d^2*x^4)/(c*d*x^2)]],-2*(1-Sqrt[2])] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c^4-a*d^4,0]
```

2: $\int \frac{c+d\,x^2}{\sqrt{a+b\,x^8}}\,dx$ when $b\,c^4-a\,d^4\neq 0$

Derivation: Algebraic expansion

- Basis: $\frac{c+d\,x^2}{\sqrt{a+b\,x^8}} == \frac{\left(d+\left(\frac{b}{a}\right)^{1/4} c\right)\left(1+\left(\frac{b}{a}\right)^{1/4} x^2\right)}{2\left(\frac{b}{a}\right)^{1/4}\sqrt{a+b\,x^8}} - \frac{\left(d-\left(\frac{b}{a}\right)^{1/4} c\right)\left(1-\left(\frac{b}{a}\right)^{1/4} x^2\right)}{2\left(\frac{b}{a}\right)^{1/4}\sqrt{a+b\,x^8}}$

Rule: If $b\,c^4-a\,d^4\neq 0$, then

$$\int \frac{c+d\,x^2}{\sqrt{a+b\,x^8}}\,dx \rightarrow \frac{d+\left(\frac{b}{a}\right)^{1/4} c}{2\left(\frac{b}{a}\right)^{1/4}}\int \frac{1+\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b\,x^8}}\,dx - \frac{d-\left(\frac{b}{a}\right)^{1/4} c}{2\left(\frac{b}{a}\right)^{1/4}}\int \frac{1-\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b\,x^8}}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)/Sqrt[a_+b_.*x_^8],x_Symbol] :=
  (d+Rt[b/a,4]*c)/(2*Rt[b/a,4])*Int[(1+Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] -
  (d-Rt[b/a,4]*c)/(2*Rt[b/a,4])*Int[(1-Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c^4-a*d^4,0]
```

5: $\int \frac{P_q[x]}{x\sqrt{a+b\,x^n}}\,dx$ when $n \in \mathbb{Z}^+ \land P_q[x,\,0] \neq 0$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+ \land P_q[x,\,0] \neq 0$, then

$$\int \frac{P_q[x]}{x\sqrt{a+b\,x^n}}\,dx \longrightarrow P_q[x,\,0]\int \frac{1}{x\sqrt{a+b\,x^n}}\,dx + \int \frac{P_q[x]-P_q[x,\,0]}{x}\,\frac{1}{\sqrt{a+b\,x^n}}\,dx$$

Program code:

```
Int[Pq_/(x_*Sqrt[a_+b_.*x_^n_]),x_Symbol] :=
  Coeff[Pq,x,0]*Int[1/(x*Sqrt[a+b*x^n]),x] +
  Int[ExpandToSum[(Pq-Coeff[Pq,x,0])/x,x]/Sqrt[a+b*x^n],x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && NeQ[Coeff[Pq,x,0],0]
```

6: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $\frac{n}{2}\in\mathbb{Z}^+\land\neg\,\text{PolynomialQ}\left[P_q[x],\,x^{\frac{n}{2}}\right]$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}^+$, then $P_q[x]=\sum_{j=0}^{n-1}x^j\sum_{k=0}^{(q-j)/n+1}P_q[x,\,j+k\,n]\,x^{k\,n}$

Note: This rule transform integrand into a sum of terms of the form $x^k\,Q_r\left[x^{\frac{n}{2}}\right]\,(a+b\,x^n)^p$.

Rule: If $\frac{n}{2}\in\mathbb{Z}^+\land\neg\,\text{PolynomialQ}\left[P_q[x],\,x^{\frac{n}{2}}\right]$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\,\rightarrow\,\int\sum_{j=0}^{\frac{n}{2}-1}x^j\left(\sum_{k=0}^{\frac{2(q-j)}{n}+1}P_q\left[x,\,j+\frac{k\,n}{2}\right]x^{\frac{k\,n}{2}}\right)(a+b\,x^n)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],j,k},
  Int[Sum[x^j*Sum[Coeff[Pq,x,j+k*n/2]*x^(k*n/2),{k,0,2*(q-j)/n+1}]*(a+b*x^n)^p,{j,0,n/2-1}],x]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && IGtQ[n/2,0] && Not[PolyQ[Pq,x^(n/2)]]
```

7: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\land q==n-1$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\land q==n-1$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\longrightarrow\;P_q[x,\,n-1]\int x^{n-1}\,(a+b\,x^n)^p\,dx+\int\left(P_q[x]-P_q[x,\,n-1]\,x^{n-1}\right)\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Coeff[Pq,x,n-1]*Int[x^(n-1)*(a+b*x^n)^p,x] +
  Int[ExpandToSum[Pq-Coeff[Pq,x,n-1]*x^(n-1),x]*(a+b*x^n)^p,x] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && IGtQ[n,0] && Expon[Pq,x]==n-1
```

8: $\int\frac{P_q[x]}{a+b\,x^n}\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int\frac{P_q[x]}{a+b\,x^n}\,dx\;\longrightarrow\;\int\text{ExpandIntegrand}\left[\frac{P_q[x]}{a+b\,x^n},\,x\right]dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^n_),x_Symbol] :=
  Int[ExpandIntegrand[Pq/(a+b*x^n),x],x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IntegerQ[n]
```

9: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\land q-n\geq 0\land q+n\,p+1\neq 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Algebraic expansion and binomial recurrence 3a

Reference: G&R 2.104

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule: If $n\in\mathbb{Z}^+\land q+n\,p+1\neq 0\land q-n\geq 0$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\longrightarrow$$

$$P_q[x,q]\int x^q\,(a+b\,x^n)^p+\int\left(P_q[x]-P_q[x,q]\,x^q\right)(a+b\,x^n)^p\,dx\,dx\;\longrightarrow$$

$$\frac{P_q[x,q]\,x^{q-n+1}\,(a+b\,x^n)^{p+1}}{b\,(q+n\,p+1)}+\frac{1}{b\,(q+n\,p+1)}\int\left(b\,(q+n\,p+1)\,\left(P_q[x]-P_q[x,q]\,x^q\right)-a\,P_q[x,q]\,(q-n+1)\,x^{q-n}\right)(a+b\,x^n)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    Pqq*x^(q-n+1)*(a+b*x^n)^(p+1)/(b*(q+n*p+1)) +
    1/(b*(q+n*p+1))*Int[ExpandToSum[b*(q+n*p+1)*(Pq-Pqq*x^q)-a*Pqq*(q-n+1)*x^(q-n),x]*(a+b*x^n)^p,x]] /;
  NeQ[q+n*p+1,0] && q-n≥0 && (IntegerQ[2*p] || IntegerQ[p+(q+1)/(2*n)])] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && IGtQ[n,0]
```

2: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $F[x] == -\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Note: $x^q\,P_q[x^{-1}]$ is a polynomial in x.

Rule: If $n\in\mathbb{Z}^-$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx\;\longrightarrow\;-\text{Subst}\left[\int\frac{x^q\,P_q[x^{-1}]\,(a+b\,x^{-n})^p}{x^{q+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  -Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x→x^(-1)],x]*(a+b*x^(-n))^p/x^(q+2),x],x,1/x]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && ILtQ[n,0]
```

5: $\int P_q[x]\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $x^m\,P_q[x]\,F[x^n] = g\,\text{Subst}\left[x^{g\,(m+1)-1}\,P_q[x^g]\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\,\partial_x\,x^{1/g}$

Rule: If $n\in\mathbb{F}$, let $g=\text{Denominator}[n]$, then

$$\int P_q[x]\,(a+b\,x^n)^p\,dx \longrightarrow g\,\text{Subst}\left[\int x^{g-1}\,P_q[x^g]\,(a+b\,x^{g\,n})^p\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g-1)*ReplaceAll[Pq,x→x^g]*(a+b*x^(g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && FractionQ[n]
```

6: $\int (A+B\,x^m)\,(a+b\,x^n)^p\,dx$ when $m-n+1=0$

Derivation: Algebraic expansion

Rule:

$$\int (A+B\,x^m)\,(a+b\,x^n)^p\,dx \longrightarrow A\int (a+b\,x^n)^p\,dx+B\int x^m\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(A_+B_.*x_^m_.)*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  A*Int[(a+b*x^n)^p,x] + B*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,A,B,m,n,p},x] && EqQ[m-n+1,0]
```

?: $\int \left(A+B\,x^{n/2}+C\,x^n+D\,x^{3n/2}\right)\left(a+b\,x^n\right)^p\,dx$ when $p+1\in\mathbb{Z}^-$

Derivation: OS and binomial recurrence

Note: This special case rule can be eliminated when there is a rule for integrands of the form $P_q[x^n]\left(a+b\,x^n+c\,x^{2n}\right)^p$.

Rule: If $p+1\in\mathbb{Z}^-$, then

$$\int \left(A+B\,x^{n/2}+C\,x^n+D\,x^{3n/2}\right)\left(a+b\,x^n\right)^p\,dx\,\longrightarrow$$
$$-\frac{x\left(b\,A-a\,C+\left(b\,B-a\,D\right)x^{n/2}\right)\left(a+b\,x^n\right)^{p+1}}{a\,b\,n\,(p+1)}-$$
$$\frac{1}{2\,a\,b\,n\,(p+1)}\int\left(a+b\,x^n\right)^{p+1}\left(2\,a\,C-2\,b\,A\,(n\,(p+1)+1)+\left(a\,D\,(n+2)-b\,B\,(n\,(2\,p+3)+2)\right)x^{n/2}\right)dx$$

Program code:

```
Int[P3_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{A=Coeff[P3,x^(n/2),0],B=Coeff[P3,x^(n/2),1],C=Coeff[P3,x^(n/2),2],D=Coeff[P3,x^(n/2),3]},
  -(x*(b*A-a*C+(b*B-a*D)*x^(n/2))*(a+b*x^n)^(p+1))/(a*b*n*(p+1)) -
  1/(2*a*b*n*(p+1))*Int[(a+b*x^n)^(p+1)*Simp[2*a*C-2*b*A*(n*(p+1)+1)+(a*D*(n+2)-b*B*(n*(2*p+3)+2))*x^(n/2),x],x]] /;
FreeQ[{a,b,n},x] && PolyQ[P3,x^(n/2),3] && ILtQ[p,-1]
```

7: $\int P_q[x]\,(a+b\,x^n)^p\,dx$

Derivation: Algebraic expansion

Rule:

$$\int P_q[x]\,(a+b\,x^n)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[P_q[x]\,(a+b\,x^n)^p,\,x\right]dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,n,p},x] && (PolyQ[Pq,x] || PolyQ[Pq,x^n])
```

S: $\int P_q[v^n]\,(a+b\,v^n)^p\,dx$ when $v == f+g\,x$

Derivation: Integration by substitution

Rule: If $v == f+g\,x$, then

$$\int P_q[v^n]\,(a+b\,v^n)^p\,dx \longrightarrow \frac{1}{g}\,\text{Subst}\left[\int P_q[x^n]\,(a+b\,x^n)^p\,dx,\,x,\,v\right]$$

Program code:

```
Int[Pq_*(a_+b_.*v_^n_.)^p_,x_Symbol] :=
  1/Coeff[v,x,1]*Subst[Int[SubstFor[v,Pq,x]*(a+b*x^n)^p,x],x,v] /;
FreeQ[{a,b,n,p},x] && LinearQ[v,x] && PolyQ[Pq,v^n]
```

Rules for integrands of the form $(h\,x)^m\,P_q[x]\,(a+b\,x^n)^p\,(c+d\,x^n)^q$

1. $\int P_q[x]\,(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p\,dx$ when $a_2\,b_1+a_1\,b_2==0$

1: $\int P_q[x]\,(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p\,dx$ when $a_2\,b_1+a_1\,b_2==0\,\land\,(p\in\mathbb{Z}\,\lor\,a_1>0\,\land\,a_2>0)$

Derivation: Algebraic simplification

Basis: If $a_2\,b_1+a_1\,b_2==0\,\land\,(p\in\mathbb{Z}\,\lor\,a_1>0\,\land\,a_2>0)$, then $(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p==(a_1\,a_2+b_1\,b_2\,x^{2\,n})^p$

Rule: If $a_2\,b_1+a_1\,b_2==0\,\land\,(p\in\mathbb{Z}\,\lor\,a_1>0\,\land\,a_2>0)$, then

$$\int P_q[x]\,(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p\,dx\,\longrightarrow\,\int P_q[x]\,(a_1\,a_2+b_1\,b_2\,x^{2\,n})^p\,dx$$

Program code:

```
Int[Pq_*(a1_+b1_.*x_^n_.)^p_.*(a2_+b2_.*x_^n_.)^p_.,x_Symbol] :=
  Int[Pq*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,n,p},x] && PolyQ[Pq,x] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

2: $\int P_q[x]\,(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p\,dx$ when $a_2\,b_1+a_1\,b_2==0$

Derivation: Piecewise constant extraction

Basis: If $a_2\,b_1+a_1\,b_2==0$, then $\partial_x\,\frac{(a_1+b_1\,x^n)^p\,(a_2+b_2\,x^n)^p}{(a_1\,a_2+b_1\,b_2\,x^{2\,n})^p}==0$

Rule: If $a_2\,b_1+a_1\,b_2==0$, then

$$\int P_q[x] \left(a_1+b_1 x^n\right)^p \left(a_2+b_2 x^n\right)^p dx \rightarrow \frac{\left(a_1+b_1 x^n\right)^{\text{FracPart}[p]} \left(a_2+b_2 x^n\right)^{\text{FracPart}[p]}}{\left(a_1 a_2+b_1 b_2 x^{2n}\right)^{\text{FracPart}[p]}} \int P_q[x] \left(a_1 a_2+b_1 b_2 x^{2n}\right)^p dx$$

Program code:

```
Int[Pq_*(a1_+b1_.*x_^n_.)^p_.*(a2_+b2_.*x_^n_.)^p_.,x_Symbol] :=
  (a1+b1*x^n)^FracPart[p]*(a2+b2*x^n)^FracPart[p]/(a1*a2+b1*b2*x^(2*n))^FracPart[p]*
    Int[Pq*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,n,p},x] && PolyQ[Pq,x] && EqQ[a2*b1+a1*b2,0] && Not[EqQ[n,1] && LinearQ[Pq,x]]
```

2: $\int \left(e+f x^n+g x^{2n}\right) \left(a+b x^n\right)^p \left(c+d x^n\right)^p dx$ when $a c f == e (b c+a d) (n (p+1)+1) \land a c g == b d e (2 n (p+1)+1)$

Rule: If $a c f == e (b c+a d) (n (p+1)+1) \land a c g == b d e (2 n (p+1)+1)$, then

$$\int \left(e+f x^n+g x^{2n}\right) \left(a+b x^n\right)^p \left(c+d x^n\right)^p dx \rightarrow \frac{e x \left(a+b x^n\right)^{p+1} \left(c+d x^n\right)^{p+1}}{a c}$$

Program code:

```
Int[(e_+f_.*x_^n_.+g_.*x_^n2_.)*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.)^p_.,x_Symbol] :=
  e*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(p+1)/(a*c) /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[n2,2*n] && EqQ[a*c*f-e*(b*c+a*d)*(n*(p+1)+1),0] && EqQ[a*c*g-b*d*e*(2*n*(p+1)+1),0]
```

```
Int[(e_+g_.*x_^n2_.)*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.)^p_.,x_Symbol] :=
  e*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(p+1)/(a*c) /;
FreeQ[{a,b,c,d,e,g,n,p},x] && EqQ[n2,2*n] && EqQ[n*(p+1)+1,0] && EqQ[a*c*g-b*d*e*(2*n*(p+1)+1),0]
```

3: $\int (A + B x^m) (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land m - n + 1 == 0$

Derivation: Algebraic expansion

Rule: If $b c - a d \neq 0 \land m - n + 1 == 0$, then

$$\int (A + B x^m) (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow A \int (a + b x^n)^p (c + d x^n)^q \, dx + B \int x^m (a + b x^n)^p (c + d x^n)^q \, dx$$

Program code:

```
Int[(A_+B_.*x_^m_.)*(a_.+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  A*Int[(a+b*x^n)^p*(c+d*x^n)^q,x] + B*Int[x^m*(a+b*x^n)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,A,B,m,n,p,q},x] && NeQ[b*c-a*d,0] && EqQ[m-n+1,0]
```

Rules for integrands of the form $P_m[x]^q \left(a+b\left(c+d\,x\right)^n\right)^p$

1: $\int P_m[x]^q \left(a+b\left(c+d\,x\right)^n\right)^p \, dx$ when $q\in\mathbb{Z}\land n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $F\left[x,\ (c+d\,x)^{1/k}\right] = \frac{k}{d}\,\text{Subst}\left[x^{k-1}\,F\left[\frac{x^k}{d}-\frac{c}{d},\ x\right],\ x,\ (c+d\,x)^{1/k}\right]\partial_x\,(c+d\,x)^{1/k}$

Rule: If $q\in\mathbb{Z}\land n\in\mathbb{F}$, let $k=\text{Denominator}[n]$, then

$$\int P_m[x]^q \left(a+b\left(c+d\,x\right)^n\right)^p dx \longrightarrow \frac{k}{d}\,\text{Subst}\left[\int x^{k-1}\,P_m\left[\frac{x^k}{d}-\frac{c}{d}\right]^q \left(a+b\,x^{k\,n}\right)^p dx,\ x,\ (c+d\,x)^{1/k}\right]$$

Program code:

```
Int[Px_^q_.*(a_.+b_.*(c_+d_.*x_)^n_)^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k/d*Subst[Int[SimplifyIntegrand[x^(k-1)*ReplaceAll[Px,x→x^k/d-c/d]^q*(a+b*x^(k*n))^p,x],x],x,(c+d*x)^(1/k)]] /;
FreeQ[{a,b,c,d,p},x] && PolynomialQ[Px,x] && IntegerQ[q] && FractionQ[n]
```